```
(*  WQ  FS  2013

  1, 2, oder 3 Wuerfel: Jeweils Augenzahl addieren.
    Zufallsvariable addieren bedeutet Verteilungsdichten falten. *)

p[n_Integer] = If[(n > 0) && (n < 7), 1./6., 0.0];
w1 = Table[6 * p[k], {k, 1, 7}]

p2[k_Integer] = Sum[p[j] * p[k - j], {j, 1, 6}];
w2 = Table[36 * p2[kk], {kk, 1, 13}]
```

```
{1., 1., 1., 1., 1., 1., 0.}

{0., 1., 2., 3., 4., 5., 6., 5., 4., 3., 2., 1., 0.}
```

```
p3[l_Integer] = Sum[p2[k] * p[l - k], {k, 1, 12}];
w3 = Table[216 * p3[k], {k, 19}]
```

```
{0., 0., 1., 3., 6., 10., 15., 21., 25., 27., 27., 25., 21., 15., 10., 6., 3., 1., 0.}
```

```
grau = GrayLevel[0.5];
hell = GrayLevel[0.8];
weiss = GrayLevel[1.0];
b1 = BarChart[{w1}, ChartStyle -> {grau}, PlotRange -> All];
b2 = BarChart[{w2}, ChartStyle -> {hell}, PlotRange -> All];
b3 = BarChart[{w3}, ChartStyle -> {weiss}, PlotRange → All,
   ChartLabels → {Range[19]}, AxesLabel → {"Augenzahl", "Anzahl"} ];

Show[b3, b2, b1]
```




```
(* Analog: The convolution of UnitBox with itself is UnitTriangle *)
Convolve[UnitBox[x], UnitBox[x], x, t]
```

```
UnitTriangle[t]
```

```
Plot[%, {t, -1.2, 1.2}]
```



```
Table[Plot[PDF[UniformSumDistribution[k], x], {x, 0, 4}, Filling → Axis,
  Exclusions → None, PlotRange → {0, 1}, PlotLabel → k], {k, 1, 4}]
```




```
(*  Gleichverteilung gegen Binomialverteilung
  6 x Wuerfel   <--> 6 x Muenzwurf
    Wuerfel habe Punkte von 0 bis 5     *)

w1 = Table[p[k], {k, 1, 6}]
b6 = Table[Binomial[5, k] / 2^5, {k, 0, 5}]
```

{0.166667, 0.166667, 0.166667, 0.166667, 0.166667, 0.166667}

$$\left\{\frac{1}{32}, \frac{5}{32}, \frac{5}{16}, \frac{5}{16}, \frac{5}{32}, \frac{1}{32}\right\}$$

```
v1 = BarChart[{w1}, ChartStyle → Opacity[0.5],
  PlotRange -> All, ChartLabels → {Range[6] - 1}]
```




```
v2 = BarChart[{b6}, ChartStyle -> {grau},
  PlotRange → All, ChartLabels → {Range[6] - 1},
  AxesLabel → {"Augenzahl/AnzahlKopf", "Anzahl"} ]
```




```
Show[v2, v1]
```



```
(*  mit Mma wirklich wuerfeln *)

w1 = Table[RandomInteger[5] + 1, {k, 1, 1000}];
w2 = Table[RandomInteger[5] + 1, {k, 1, 1000}];
w3 = Table[RandomInteger[5] + 1, {k, 1, 1000}];

Short[w1]

{3, 1, 5, 3, 4, 5, 3, 4, 1, 5, 1, 5, 4, 4,
 3, <<970>>, 4, 3, 2, 1, 5, 4, 5, 4, 6, 4, 2, 4, 4, 4, 6}

Histogram[w1]
```




```
Histogram[w1 + w2]
```



```
Histogram[w1 + w2 + w3]
```




```
(* Mma kennt alle Verteilungen, hier die Normalverteilung *)
```

```
da1 = Histogram[RandomVariate[NormalDistribution[0, 1], 1000]]
```




```
da2 = Histogram[RandomVariate[NormalDistribution[3, 1 / 2], 1000]]
```



```
Show[da2, da1, PlotRange → All]
```




```
(* Overlay several probability density
 function ( PDF ) plots for Poisson distributions: *)

PDF[PoissonDistribution[λ], k]
```

$$\begin{cases} \frac{e^{-\lambda}\lambda^k}{k!} & k \geq 0 \\ 0 & \text{True} \end{cases}$$

```
Show[Histogram[Table[RandomInteger[PoissonDistribution[10 i], 1000], {i, 5}],
  {1}, "PDF", PlotRange → All],
 Plot[Evaluate[Table[PDF[PoissonDistribution[10 μ], x], {μ, 5}]],
  {x, 0, 80}, PlotStyle → Thick, PlotRange → All]]
```




```
(* PDF direkt *)

PDF[NormalDistribution[0, 1], x] (*  my=0, sigma=1 *)
```

$$\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\pi}}$$

```
PDF[NormalDistribution[3, 1 / 2], x] (*  my=3, sigma=1/2 *)
```

$$e^{-2(-3+x)^2}\sqrt{\frac{2}{\pi}}$$

```
Plot[PDF[NormalDistribution[0, 1], x], {x, -5, 5}, Filling → Axis]
```




```
punkte = Graphics[{PointSize[0.02],
    Table[Point[{x, N[PDF[NormalDistribution[0, 1], x]]}], {x, 1, 3}]}];
bi = Plot[{PDF[NormalDistribution[0, 1], x] , PDF[NormalDistribution[0, 2], x],
    PDF[NormalDistribution[0, 3], x]}, {x, -3, 7}];
Show[
 bi,
 punkte]
```




```
(*  Die Markierungen sind bei 1 sigma, 2 sigma und 3 sigma
    entsprechend  2/3,  95%,  und  99.5%  der Flaeche *)


(* Noch mal Faltung *)

p[n_Integer] = If[(n > 0) && (n < 7), 1. / 6., 0.0];
```

```
DiscretePlot[p[n], {n, 0, 6}]
```




```
DiscreteConvolve[p[n], p[n], n, m]
```

```
0.          m > 12 || m < 2
0.0277778   m == 2 || m == 12.
0.0555556   m == 3 || m == 11.
0.0833333   m == 4 || m == 10.
0.111111    m == 5 || m == 9.
0.138889    m == 6 || m == 8.
0.166667    True
```

```
DiscretePlot[%, {m, 0, 13}]
```




```
(* Faltung fuer stetige Dichten  *)
```

```
Convolve[PDF[NormalDistribution[3, 1 / 2], x],
 PDF[NormalDistribution[0, 1], x], x, y]
```

$$e^{-\frac{2}{5}(-3+y)^2}\sqrt{\frac{2}{5\pi}}$$

```
(*  Fuer die Normalverteilung gilt:
   Ihre Faltung mit sich selbt gibt wieder eine Normalverteilung mit
     my= my_1 + my_2 und sigma^2=sigma_1^2 + sigma_2^2 *)
```

```
sigma = Sqrt[1 / 4 + 1] ;  PDF[NormalDistribution[3, Sqrt[5] / 2], x]
```

$$e^{-\frac{2}{5}(-3+x)^2}\sqrt{\frac{2}{5\pi}}$$

```
(*  Eine analoge Relation gilt fuer die Binomialverteilung *)

PDF[BinomialDistribution[n, p], k]
```

$$\begin{cases}(1-p)^{-k+n}\,p^k\,\text{Binomial}[n,k] & 0\le k\le n\\ 0 & \text{True}\end{cases}$$

```
Mean[BinomialDistribution[n, p]]
Variance[BinomialDistribution[n, p]]
```

n p

n (1 - p) p

```
DiscretePlot[
 Evaluate@Table[PDF[BinomialDistribution[40, p], k], {p, {0.1, 0.5, 0.7}}],
 {k, 40}, PlotRange → All, PlotMarkers → Automatic]
```




```
DiscretePlot[CDF[BinomialDistribution[40, .7], x],
 {x, 10, 40}, ExtentSize → Right, ExtentMarkers → {"Filled", "Empty"}]
```



```
CDF[BinomialDistribution[5, .6], x] // PiecewiseExpand
```

$$\begin{cases} 0.01024 & 0 \le x < 1 \\ 0.08704 & 1 \le x < 2 \\ 0.31744 & 2 \le x < 3 \\ 0.66304 & 3 \le x < 4 \\ 0.92224 & 4 \le x < 5 \\ 1 & x > 5 \\ 1. & x == 5 \\ 0 & \text{True} \end{cases}$$

```
RandomVariate[BernoulliDistribution[0.75], 10]
{1, 1, 0, 1, 1, 1, 0, 1, 1, 1}

(*  Die Wahrscheinlichkeit eines Tores beim Elfmeter *)

In[1]:= RandomVariate[BernoulliDistribution[0.8], 10] /. {0 → "Gehalten", 1 → "Tor"}
Out[1]= {Tor, Tor, Gehalten, Tor, Tor, Tor, Gehalten, Tor, Tor, Tor}

(*A packet consisting of a string of n symbols is
  transmitted over a noisy channel. Each symbol has probability
  10^-4 of incorrect transmission. Find  n such that the probability
  of incorrect packet transmission is less than 10^-3: *)
Probability[x ≥ 1, x  BinomialDistribution[n, 10^-4]]
```

$$1 - \left(\frac{9999}{10\,000}\right)^n$$

```
Reduce[% ≤ 10^-3, n, Reals] // N
n ≤ 10.0045

Compute the same limit using a Poisson approximation :

Probability[x ≥ 1, x  PoissonDistribution[n 10^-4]]
```

$$1 - e^{-n/10\,000}$$

```
Reduce[% ≤ 10^-3, n, Reals] // N
n ≤ 10.005

(*  Faltung von Binomial Dichten *)

PDF[BinomialDistribution[n, p], k]
```

$$\begin{cases} (1-p)^{-k+n} p^k \text{Binomial}[n, k] & 0 \le k \le n \\ 0 & \text{True} \end{cases}$$

```
bi1[n_Integer, k_Integer, p_Real] =
  If[(n >= 0) && (k >= 0) && (k <= n), (1 - p)^(-k+n) p^k Binomial[n, k] , 0.0];

(* DiscreteConvolve[bi1[10,k,0.7],bi1[20,k,0.7],k,r]
   geht nicht ?? Nutze direkte Faltungsformel *)

convBi[r_] := Sum[bi1[28, k, 0.7] * bi1[22, r - k, 0.7], {k, 0, 50}];
```

```
DiscretePlot[convBi[r], {r, 20, 50}]
```




```
(*  Vergleich der beiden Faelle: ist natuerlich kein Beweis! *)

DiscretePlot[PDF[BinomialDistribution[50, 0.7], k], {k, 20, 50}]
```




```
(* Wenn x≃BinomialDistribution[n,p] und y≃BinomialDistribution[m,p]
  verteilt sind, dann ist eine Zufallsgroesse z=x+y nach
    z≃BinomialDistribution[n+m,p] verteilt *)


(*  ErlangDistribution *)

(* Die ErlangDistribution[k,λ] is the
 convolution of  k ExponentialDistribution[λ]  PDFs: *)

Convolve[PDF[ErlangDistribution[k, λ], x],
 PDF[ExponentialDistribution[λ], x], x, y]
```

$$\begin{cases} \frac{e^{-y\lambda}\,\lambda\,(y\,\lambda)^{k}}{k\,\text{Gamma}[k]} & y > 0 \\ 0 & \text{True} \end{cases}$$

```
PDF[ErlangDistribution[k + 1, λ], y]
```

$$\begin{cases} \frac{e^{-y\lambda}\,y^{k}\,\lambda^{1+k}}{\text{Gamma}[1+k]} & y > 0 \\ 0 & \text{True} \end{cases}$$

```
Table[Plot[PDF[ErlangDistribution[k, 1], x], {x, 0, 8}, Filling → Axis,
  Exclusions → None, PlotRange → {0, 1}, PlotLabel → k], {k, 1, 4}]
```